Shaving with Occam’s Razor: Deriving Minimalist Theorem Provers for Minimal Logic

Paul Tarau

University of North Texas

RCRA’2018

Research supported by NSF grant 1423324
Outline

1. The implicational fragment of propositional intuitionistic logic
2. Proof systems for intuitionistic implicational propositional logic
3. An executable specification
4. Deriving our *lean* theorem provers
5. The testing framework
6. Conclusions and future work

code is available at: https://github.com/ptarau/TypesAndProofs
The implicational fragment of propositional intuitionistic logic
Hilbert-style axioms schemes for the implicational fragment of propositional intuitionistic logic

the implicational fragment of intuitionistic propositional logic can be defined by two axiom schemes:

- \( K : \ A \rightarrow (B \rightarrow A) \)
- \( S : \ (A \rightarrow (B \rightarrow C)) \rightarrow ((A \rightarrow B) \rightarrow (A \rightarrow C)) \)

and the modus ponens inference rule:

- \( MP : \ A, A \rightarrow B \vdash B \).

The insight: those are exactly the types of the combinators \( S \) and \( K \)!

Is there a bridge standing up between the two sides?
The bridge between **types** and **propositions**

Curry-Howard isomorphism
The Curry-Howard isomorphism

it connects:

- the implicational fragment of propositional intuitionistic logic
- types in the *simply typed lambda calculus*

complexity of “crossing the bridge”, different in the two directions

- a (low polynomial) type inference algorithm associates a type (when it exists) to a lambda term
- PSPACE-complete algorithms associate lambda terms as inhabitants to a given type expression

⇒

- lambda term (typically in normal form) can serve as a witness for the existence of a proof for the corresponding tautology in minimal logic
- a theorem prover can also be seen as a tool for program synthesis
Proof systems for intuitionistic implicational propositional logic
Gentzen’s \( \text{LJ} \) calculus, reduced to the implicational fragment of intuitionistic propositional logic

- \( \text{LJ}_1 \): 
  \[
  A, \Gamma \vdash A
  \]

- \( \text{LJ}_2 \): 
  \[
  A, \Gamma \vdash B \\
  \Gamma \vdash A \rightarrow B \\
  \Gamma \vdash A \rightarrow B
  \]

- \( \text{LJ}_3 \): 
  \[
  A \rightarrow B, \Gamma \vdash A \\
  B, \Gamma \vdash G \\
  A \rightarrow B, \Gamma \vdash G
  \]

- Rules, if implemented directly are subject to looping
- Several variants use loop-checking, by recording the sequents used
Dyckhoff’s **LJT** calculus (implicational fragment)

- replace $LJ_3$ with $LJT_3$ and $LJT_4$
- termination proven using multiset orderings
- no need for loop checking
- efficient and simple

- **$LJT_1$** : \[ A, \Gamma \vdash A \]
- **$LJT_2$** : \[ A, \Gamma \vdash B \quad \Gamma \vdash A \rightarrow B \]
- **$LJT_3$** : \[ B, A, \Gamma \vdash G \quad \quad \quad [A \text{ atomic }] \]
  \[ A \rightarrow B, A, \Gamma \vdash G \]
- **$LJT_4$** : \[ D \rightarrow B, \Gamma \vdash C \rightarrow D \quad B, \Gamma \rightarrow G \]
  \[ (C \rightarrow D) \rightarrow B, \Gamma \vdash G \]

To support negation, a rule for the special term *false* is needed

- **$LJT_5$** : \[ false, \Gamma \vdash G \]
An executable specification
Notations and assumptions

- we use **Prolog** as our meta-language
- code (now grown to above 4000 lines, covering full propositional logic) at [https://github.com/ptarau/TypesAndProofs](https://github.com/ptarau/TypesAndProofs)
- Prolog programming background:
  - variables will be denoted with uppercase letters
  - the pure Horn clause subset
  - well-known built-in predicates like `memberchk/2` and `select/3`, `call/N`
  - **CUT** and **if-then-else** constructs
- **lambda terms**: `a/2` = application, `l/2` = lambda binders with a variable as its first argument, an expression as second and **logic variables** representing the leaf variables bound by a lambda
- **type expressions** (also seen as implicational formulas): binary trees with the function symbol “`->/2`” and **logic variables (or atoms or integers)** as their leaves
Examples

the S combinator and its type, with variables and integers as leaves:
The importance of being Leanest

- Roy Dyckchoff’s program, about 420 lines
- can we just use his calculus as a starting point?
- a blast from the past: lean theorem provers can be fast!

⇒

- we start with a simple, almost literal translation of rules $LJT_1 \ldots LJT_4$ to Prolog
- note: values in the environment $\Gamma$ denoted by the variables $Vs, Vs_1, Vs_2 \ldots$. 
Dyckhoff’s LJT calculus, literally

lprove(T):-ljt(T,[]),!.

ljt(A, Vs):-memberchk(A, Vs),!.

ljt((A->B), Vs):-!,ljt(B, [A|Vs]).

ljt(G, Vs1):- % atomic(G),
    select((A->B), Vs1, Vs2),
    atomic(A),
    memberchk(A, Vs2),
    !,
    ljt(G, [B|Vs2]).

ljt(G, Vs1):- % atomic(G),
    select(((C->D)->B), Vs1, Vs2),
    ljt((C->D), [(D->B)|Vs2]),
    !,
    ljt(G, [B|Vs2]).
Deriving our *lean* theorem provers
bprove: concentrating nondeterminism into one place

The first transformation merges the work of the two `select/3` calls into a single call, observing that they do similar things after the call. That avoids redoing the same iteration over candidates for reduction.

```prolog
bprove(T) :- ljb(T, []), !.

ljb(A, Vs) :- memberchk(A, Vs), !.
ljb((A->B), Vs) :- !, ljb(B, [A|Vs]).
ljb(G, Vs1) :-
    select((A->B), Vs1, Vs2),
    ljb_imp(A, B, Vs2),
    !,
    ljb(G, [B|Vs2]).

ljb_imp((C->D), B, Vs) :- !, ljb((C->D), [ (D->B) | Vs]).
ljb_imp(A, _, Vs) :- atomic(A), memberchk(A, Vs).
```

⇒ 51% speed improvement for formulas with 14 internal nodes
sprove: extracting the proof terms

sprove(T,X):-ljs(X,T,[]),!.

ljs(X,A,Vs):-memberchk(X:A,Vs),!.
   % leaf variable
ljs(l(X,E),(A->B),Vs):-!,ljs(E,B,[X:A|Vs]).  % lambda term
ljs(E,G,Vs1):-
  member(_:V,Vs1),head_of(V,G),!, % fail if non-tautology
  select(S:(A->B),Vs1,Vs2), % source of application
  ljs_imp(T,A,B,Vs2), % target of application
  !,
  ljs(E,G,[a(S,T):B|Vs2]). % application

ljs_imp(E,A,_,Vs):-atomic(A),!,memberchk(E:A,Vs).
ljs_imp(l(X,l(Y,E)),(C->D),B,Vs):-ljs(E,D,[X:C,Y:(D->B)|Vs]).

head_of(_->B,G):-!,head_of(B,G).
head_of(G,G).
Extracting **S**, **K** and **I** from their types

?- sprove((0→1→2)→(0→1)→0→2),X).
X = l(A, l(B, l(C, a(a(A, C), a(B, C)))))). % S

?- sprove((0→1→0),X).
X = l(A, l(B, A)). % K

?- sprove((0→0),X).
X = l(A, A). % I
Steps for inferring $S$ from its type

?- s__(S),sprove(S,X).

[] --> A: ((0→1→2)→(0→1)→0→2)  
[A: (0→1→2)] --> B: ((0→1)→0→2)  
[A: (0→1), B: (0→1→2)] --> C: (0→2)  
[A: 0, B: (0→1), C: (0→1→2)] --> D: 2  
[a(A,B):1, B: 0, C: (0→1→2)] --> D: 2  
[a(A,B): (1→2), a(C,B):1, B: 0] --> D: 2  
[a(a(A,B), a(C,B)):2, a(C,B):1, B: 0] --> D: 2

$S = ((0→1→2)→(0→1)→0→2),  
X = l(A, l(B, l(C, a(a(A,C), a(B,C))))).$
Implicational formulas as embedded Horn Clauses

- equivalence between:
  - \( B_1 \rightarrow B_2 \rightarrow \ldots \rightarrow B_n \rightarrow H \) and
  - \( H :\neg B_1, B_2, \ldots, B_n \) (in Prolog notation)

- \( H \) is the \textit{atomic} formula ending a chain of implications

- we can recursively transform an implicational formula:

\[
\begin{align*}
toHorn & (\neg A \rightarrow B, H :- Bs) :\neg!, toHorns (\neg A \rightarrow B, Bs, H) . \\
toHorn & (H, H).
\end{align*}
\]

\[
\begin{align*}
toHorns & (\neg A \rightarrow B, [HA | Bs], H) :\neg!, toHorn (A, HA), toHorns (B, Bs, H) . \\
toHorns & (H, [], H).
\end{align*}
\]

- the transformation is reversible!

\[
\begin{align*}
?- toHorn ((0 \rightarrow 1 \rightarrow 2) \rightarrow (0 \rightarrow 1) \rightarrow 0 \rightarrow 2), R) . \\
R & = (2 :\neg [(2 :\neg [0, 1]), (1 :\neg [0]), 0]).
\end{align*}
\]

\[
\begin{align*}
?- toHorn ((0 \rightarrow 1 \rightarrow 2 \rightarrow 3 \rightarrow 4) \rightarrow (0 \rightarrow 1 \rightarrow 2) \rightarrow 0 \rightarrow 2 \rightarrow 3), R) . \\
R & = (3 :\neg [(4 :\neg [0, 1, 2, 3]), (2 :\neg [0, 1]), 0, 2]).
\end{align*}
\]
Transforming provers for implicational formulas into equivalent provers working on embedded Horn clauses

hprove(T0):-toHorn(T0,T),ljh(T,[]),!.

ljh(A,Vs):-memberchk(A,Vs),!.
ljh((B:-As),Vs1):-!,append(As,Vs1,Vs2),ljh(B,Vs2).
ljh(G,Vs1):- % atomic(G), G not on Vs1
  memberchk((G:-_),Vs1), % if non-tautology, we just fail
  select((B:-As),Vs1,Vs2), % outer select loop
  select(A,As,Bs), % inner select loop
  ljh_imp(A,B,Vs2), % A is in the body of B
  !,trimmed((B:-Bs),NewB), % trim empty bodies
  ljh(G,[NewB|Vs2]).

ljh_imp(A,_B,Vs):-atomic(A),!,memberchk(A,Vs).
ljh_imp((D:-Cs),B,Vs):-ljh((D:-Cs),[(B:-[D])|Vs]).

trimmed((B:-[]),R):-!,R=B.
trimmed(BBs,BBs).
What’s new with the embedded Horn clause form?

The embedded Horn clause form helps bypassing some intermediate steps, by focusing on the head of the Horn clause, which corresponds to the last atom in a chain of implications. Also, 69% faster on terms of size 15.

- the 3-rd clause of ljh works as a context reducer
- a second select/3 call in it gives ljh_imp more chances to succeed and commit
- it removes a clause \( B : \neg A_s \) and it removes from its body \( A_s \) a formula \( A \), to be passed to ljh_imp, with the remaining context
- if \( A \) is atomic, we succeed if and only if it is already in the context
- we closely mimic rule \( LJT_4 \) by trying to prove \( A = (D : \neg Cs) \), after extending the context with the assumption \( B : \neg [D] \).
- but here we relate \( D \) with the head \( B \)!
- the context gets smaller as \( A_s \) does not contain the \( A \) anymore
- if the body \( B_s \) is empty, the clause is downgraded to its head
A lifting to classical logic, via Glivenko’s transformation

Glivenko’s translation that prefixes a formula with its double negation. It turns an intuitionistic propositional prover into a classical one.

- we add the atom \( \text{false} \), to the language of the formulas
- we rewrite negation of \( x \) into \( x \rightarrow \text{false} \)
- we add the special handling of \( \text{false} \) as the first clause of the predicate \( \text{ljb}/2 \), corresponding to rule

\[
LJT_5 : \quad \frac{\text{false}, \Gamma \models G}{\text{ljb}(_{\text{AnyGoal}}, \text{Vs}) : \text{-memberchk(false, Vs)}, !.}
\]
The testing framework
Combinatorial testing, automated

- testing correctness:
  - a false positive: it is not a tautology, but the prover proves it
  - a false negative: it is a tautology but the prover fails on it
  - no false positive: a prover is sound
  - no false negative: a prover is complete
  - indirect testing: via Glivenko’s translation
  - soundness and completeness are relative to a "gold standard"!

- helpers:
  - intuitionistic tautologies are also classical, so if it is not classical it cannot be intuitionistic
  - crossing the Curry-Howard bridge: types of all lambda terms up to a given size: types of simply typed lambda terms are tautologies for sure

- exhaustive vs. random
  - all implicational formulas up to given size: a mix of non-tautologies and tautologies (fewer and fewer with size)
  - type of all lambda terms of a given size, random simply typed terms
  - random simply typed lambda terms, random implicational formulas
Testing against a trusted reference implementation

Once we can trust an existing reference implementation (e.g., after it passes our generator-based tests), it makes sense to use it as a gold standard. Thus, we can identify both false positives and negatives directly!

gold_test(N,Generator,Gold,Silver, Term, Res):-
    call(Generator,N,Term),
    gold_test_one(Gold,Silver,Term, Res),
    Res\=agreement.

gold_test_one(Gold,Silver,T, Res):-
    ( call(Silver,T) -> \+ call(Gold,T),
        Res = wrong_success
    ; call(Gold,T) -> % \+ Silver
        Res = wrong_failure
    ; Res = agreement
    ).
Random implicational formulas from binary trees and set partitions

The combined generator, produces in a few seconds terms of size 1000:

?- ranImpFormula(20,F).
F = (((0->(((1->2)->1->2->2)->3)->2)->4->(3->3)->
(5->2)->6->3)->7->(4->5)->(4->8)->8) .

?- time(ranImpFormula(1000,_)).
% includes tabling large Stirling numbers
% 37,245,709 inferences, 7.501 CPU in
7.975 seconds (94% CPU, 4965628 Lips)

?- time(ranImpFormula(1000,_)). % fast, thanks to tabling
% 107,163 inferences, 0.040 CPU in
0.044 seconds (92% CPU, 2659329 Lips)

superexponential growth with N, Catalan(N)*Bell(N+1)
Can our lean provers actually be fast? A quick performance evaluation

- **our benchmarking code is at**: https://github.com/ptarau/TypesAndProofs/blob/master/bm.pro.
- **we compare our provers on:**
  - known tautologies with given proof size $N$ (lambda terms in normal forms)
  - implicational formulas of size $(N/\sqrt{2})$
  - for the winner, we also test it on larger formulas up to size 20 and 10
### Runtimes on known tautologies and all formulas

<table>
<thead>
<tr>
<th>Prover</th>
<th>Term Size</th>
<th>Positive</th>
<th>Mix (half size)</th>
<th>Total seconds</th>
</tr>
</thead>
<tbody>
<tr>
<td>lprove</td>
<td>13</td>
<td>1.4</td>
<td>0.28</td>
<td>1.68</td>
</tr>
<tr>
<td>lprove</td>
<td>14</td>
<td>6.86</td>
<td>6.33</td>
<td>13.2</td>
</tr>
<tr>
<td>lprove</td>
<td>15</td>
<td>56.93</td>
<td>6.56</td>
<td><strong>63.49</strong></td>
</tr>
<tr>
<td>bprove</td>
<td>13</td>
<td>0.92</td>
<td>0.20</td>
<td>1.12</td>
</tr>
<tr>
<td>bprove</td>
<td>14</td>
<td>4.31</td>
<td>4.26</td>
<td>8.58</td>
</tr>
<tr>
<td>bprove</td>
<td>15</td>
<td>31.72</td>
<td>4.31</td>
<td><strong>36.03</strong></td>
</tr>
<tr>
<td>sprove</td>
<td>13</td>
<td>1.92</td>
<td>0.16</td>
<td>2.09</td>
</tr>
<tr>
<td>sprove</td>
<td>14</td>
<td>9.43</td>
<td>2.72</td>
<td>12.16</td>
</tr>
<tr>
<td>sprove</td>
<td>15</td>
<td>48.55</td>
<td>2.73</td>
<td><strong>51.29</strong></td>
</tr>
<tr>
<td>hprove</td>
<td>13</td>
<td>0.95</td>
<td>0.11</td>
<td>1.07</td>
</tr>
<tr>
<td>hprove</td>
<td>14</td>
<td>4.26</td>
<td>1.86</td>
<td>6.12</td>
</tr>
<tr>
<td>hprove</td>
<td>15</td>
<td>19.35</td>
<td>1.87</td>
<td><strong>21.22</strong></td>
</tr>
<tr>
<td>dprove</td>
<td>13</td>
<td>2.18</td>
<td>0.35</td>
<td>2.53</td>
</tr>
<tr>
<td>dprove</td>
<td>14</td>
<td>10.96</td>
<td>6.25</td>
<td>17.21</td>
</tr>
<tr>
<td>dprove</td>
<td>15</td>
<td>1100.72</td>
<td>5.76</td>
<td><strong>1106.49</strong></td>
</tr>
</tbody>
</table>
### How does `hprove/1` scale?

<table>
<thead>
<tr>
<th>Prover</th>
<th>Size</th>
<th>Positive</th>
<th>Mix (half-size)</th>
<th>Total Time</th>
</tr>
</thead>
<tbody>
<tr>
<td>hprove</td>
<td>16</td>
<td>89.58</td>
<td>34.74</td>
<td>124.32</td>
</tr>
<tr>
<td>hprove</td>
<td>17</td>
<td>427.47</td>
<td>33.56</td>
<td>461.03</td>
</tr>
<tr>
<td>hprove</td>
<td>18</td>
<td>2090.77</td>
<td>684.15</td>
<td>2774.92</td>
</tr>
<tr>
<td>hprove</td>
<td>19</td>
<td>11270.35</td>
<td>756.8</td>
<td><strong>12027.15</strong></td>
</tr>
</tbody>
</table>

Figure: `hprove/1` on larger tests, time in seconds

- no unexpected slowdown on either proving known tautologies or rejecting non-tautologies (contrary to `dprove/1` that gave up already on size 15)
- small enough to be converted to C, possibly competitive with much more complex provers
- needs to be tested on hard, human-made formulas (e.g., those known to have exponentially long proofs)
Related work I

- The related work derived from Gentzen’s $\text{LJ}$ calculus is in the hundreds if not in the thousands of papers and books.

- [1, 2]: Our starting points for deriving our provers, directly from the $\text{LJT}$ calculus.

- Similar calculi, key ideas of which made it into the Coq proof assistant’s code: in [3].

- [4] Described in full detail in [5], finds and/or counts inhabitants of simple types in long normal form.

- Interestingly, these algorithms have not crossed, at our best knowledge, to the other side of the Curry-Howard isomorphism in the form of theorem provers.
Related work II

- overviews of closely related calculi, using the implicational subset of propositional intuitionistic logic are [6, 2].
- using hypothetical implications in Prolog, although all with a different semantics than Gentzen’s LJ calculus or its LJT variant, go back as early as [7], followed by a series of Lambda-Prolog and linear logic-related books and papers, e.g., [8]
- the similarity to the propositional subsets of N-Prolog [7] and λ-Prolog [8] comes from their close connection to intuitionistic logic
- but neither derive implementations from a pure LJ-based calculus or have termination properties implemented along the lines the LJT calculus
Dyckhoff, R.:
*Contraction-free sequent calculi for intuitionistic logic.*

Dyckhoff, R.:
*Intuitionistic Decision Procedures Since Gentzen.*

Herbelin, H.:
*A Lambda-Calculus Structure Isomorphic to Gentzen-Style Sequent Calculus Structure.*

Ben-Yelles, C.B.:
*Type assignment in the lambda-calculus: Syntax and semantics.*
PhD thesis, University College of Swansea (1979)

Hindley, J.R.:
*Basic Simple Type Theory.*

Gabbay, D., Olivetti, N.:
*Goal-oriented deductions.*
In: *Handbook of Philosophical Logic.*

Gabbay, D.M., Reyle, U.:
*N-prolog: An extension of prolog with hypothetical implications. i.*

Miller, D., Nadathur, G.:
*Programming with Higher-Order Logic.*
Cambridge University Press, New York, NY, USA (2012)
Conclusions and future work
Conclusions and future work

- our empirically oriented approach has found variants of lean propositional intuitionistic provers that are comparable to their more complex peers, derived from similar calculi.
- besides the derivation of our lean theorem provers, our code base at https://github.com/ptarau/TypesAndProofs also provides an extensive test-driven development framework built on several cross-testing opportunities between type inference algorithms for lambda terms and theorem provers for propositional intuitionistic logic.
- the embedded Horn clause provers might be worth formalizing as a calculus and subject to deeper theoretical analysis.
- extension to full propositional and first order intuitionistic logic seems easy.
- given that they share their main data structures with Prolog, it seems interesting to attempt their partial evaluation or compilation to Prolog.
Questions?